
! Modeling a Heat Exchanger.
   A heat exchanger is a device for transfering heat from a hot fluid to a cold fluid.
Examples are: various hot and cold fluids in an oil refinery or power plant,
 a car radiator, air conditioning coils, or a gas powered furnace.

  Keywords: Heat exchanger, LINGO, LMTD, Log Mean Temperature Difference, Petroleum, 
Refinery;

! Basic idea:
   The amount of heat transfered in a small section of a heat exchanger is proportional to
 (temperature difference between the two fluids)*(heat conductance of the separating 
material).
If we can define a mean temperature difference for the entire area or length
of the heat exchanger, the total heat transferred per unit of time is proportional to 
   Area*(heat conductance of the separating material)*(average temperature difference 
between the two fluids).
Define:
  THI = temperature of the hot fluid at its inlet,
  THO = temperature of the hot fluid at its outlet,
  TCI = temperature of the cold fluid at its inlet,
  TCO = temperature of the cold fluid at its outlet.
  HMASS = mass of hot fluid per unit time through exchanger,
  CMASS = mass of cold fluid per unit time through exchanger,
  HSPH = specific heat content of hot fluid in energy/(mass*temp_change),
  CSPH = specific heat content of the cold fluid;

SUBMODEL HeatXchng:
! Define:
  THI = temperature of the hot fluid at its inlet,
  THO = temperature of the hot fluid at its outlet,
  TCI = temperature of the cold fluid at its inlet,
  TCO = temperature of the cold fluid at its outlet.
  HMASS = mass of hot fluid per unit time through exchanger,
  CMASS = mass of cold fluid per unit time through exchanger,
  HSPH = specific heat content of hot fluid in energy/(mass*temp_change),
  CSPH = specific heat content of the cold fluid,
! Case 1;
! 100 liters per minute of water needs to be cooled from 90 C to 65 C,
  The cold fluid is water with an input temperature of 20 C.
  The area of the heat exchanger is 4 m^2. The conductance is
  1000 W/m^2*degrees C). Note that water is less dense at higher temperature.
  How much cold water is needed? What will be its outlet temperature?;
!Ex01   THI = 90;     ! Hot fluid  input temp in degrees C;
!Ex01   THO = 65;     ! Hot fluid output temp in degrees C;
!Ex01   TCI = 20;     ! Cold fluid input in degrees C;
!Ex01   U = 1000;     ! Heat transfer conductance in Joules/(sec*meter^2*degreesC);
!Ex01   A = 4;        ! A = Area in square meters;
!Ex01   HSPH = 4191; ! Specific heat of hot fluid in Joules/(kg*DegreesC);
!Ex01   CSPH = 4175; ! Specific heat of cold fluid in Joules/(kg*degreesC);
!Ex01   HMASS = 1.618333; ! kg/s of hot fluid to be cooled = 100*0.971/60;
!Ex01 !  CMASS = ??; ! kg/s of cold fluid to be heated;
!Ex01 !  TCO = ?? (50.1);    ! Cold fluid is warmed up to ?, in degrees C;

! Example 2. Cool some fuel oil, What area is required and
     how much cooling water is needed? ;
!Ex02  THI = 185;     ! Hot fluid input temp in degrees F;
!Ex02  THO = 140;     ! Hot fluid output temp,must be cooled to, in degrees F;
!Ex02  TCI = 50;      ! Cold fluid input in degrees F;
!Ex02  TCO = 90;      ! Cold fluid output, is warmed up to, in degrees F;
!Ex02  U = 120;       ! Heat transfer conductance in BTU/(hr*sqfoot*degreesF),
                         sometimes called k value;
!Ex02  HSPH = 0.74;   ! Specific heat of hot fluid in BTU/(lb*DegreesFahrenheit);
!Ex02  CSPH = 1;      ! Specific heat of cold fluid in BTU/(lb*degreesF);
!Ex02  HMASS = 50000; ! lb/hr of hot fluid to be cooled;
!Ex02  ! CMASS = ??; ! lb/hr of cold fluid to be heated;
!Ex02  ! A = ??;       ! A = Area in square feet;

! Example 3. Cool some fuel oil, from 185 to 140. We can do this by either



   using a lot of cooling water or an exchanger with large area.
   Which combination should we use? ;
!Ex03; THI = 185;     ! Hot fluid input temp in degrees F;
!Ex03; THO = 140;     ! Hot fluid output temp,must be cooled to, in degrees F;
!Ex03; TCI = 50;      ! Cold fluid input in degrees F;
!Ex03; !TCO = ??;      ! Cold fluid output, is warmed up to, in degrees F;
!Ex03; U = 120;       ! Heat transfer conductance in BTU/(hr*sqfoot*degreesF),
                         sometimes called k value;
!Ex03; HSPH = 0.74;   ! Specific heat of hot fluid in BTU/(lb*DegreesFahrenheit);
!Ex03; CSPH = 1;      ! Specific heat of cold fluid in BTU/(lb*degreesF);
!Ex03; HMASS = 50000; ! lb/hr of hot fluid to be cooled;
!Ex03; ACOST = 20; ! Relative cost of heat exchanger size;
!Ex03; CCOST = 1;  ! Relative cost of cooling water;
!Ex03; CMASS <= 43000; ! Cannot use too much cooling water;
!Ex03; ! CMASS = ??; ! lb/hr of cold fluid to be heated;
!Ex03; ! A = ??;       ! A = Area in square feet;
!Ex03; A <= 155;   ! Cannot use too big a heat exchanger;
!EX03; TCO <= 95;  ! Cold water cannot be heated to high;

 MIN = OBJ;
   OBJ = ACOST * A + CCOST * CMASS; ! Minimize combined cost;

!For a counter flow heat exhanger, define;
  DTH = THI - TCO ; ! temperature difference at the hot end;
  DTC = THO - TCI ; ! temperature difference at the cold end,
! For a parallel flow heat exhanger, define;
!  DTH = THI - TCI ; ! temperature difference at the hot end;
!  DTC = THO - TCO ; ! temperature difference at the cold end;

! A simple estimate of the average temperature diffence is (DTH + DTC)/2.
It has been found that for heat exhangers, a better estimate of the 
average temperature difference is the Logarithmic Mean;
!   MTD = (DTH - DTC)/@LOG( DTH/ DTC);
! Now LINGO (and What'sBest!) have a built-in function, @LMTD, to represent this;
    MTD = @LMTD( DTH, DTC);  ! If we use the @LMTD function;
   
! Quantity of heat transferred per unit time is the product 
 of the conductivity * area * mean temperature difference;
   Q = U * A * MTD;

! Quantity of heat lost per unit time by the hot fluid = 
   mass of fluid per time * specific heat * temperature loss;
   Q = HMASS * HSPH * (THI - THO);

! Quantity of heat gained per unit time by the cold fluid;
   Q = CMASS * CSPH * ( TCO - TCI);
ENDSUBMODEL

CALC:
! Set some parameters before the solve;
  @SET( 'TERSEO',1); ! Output level (0:verbose, 1:terse, 2:only errors, 3:none);
  @SET( 'IPTOLR', 0.0001);! Set IP ending relative optimality tolerance(Should be >0);
  @SET( 'TATSLV', 150); ! Solver time limit in seconds (0:no limit) for @SOLVE's;

  @SOLVE( HeatXchng);  ! Solve a specified submodel;
  ISTAT = @STATUS();! 0: Optimal to tolerance. 1: infeasible, 2: unbounded,
                      3: undetermined, 4: Feasible,  5: Infeasible/unbounded in 
preprocessor,
                      6: Local optimum, 7: locally infeasible, 8: Objective cutoff 
reached,
                      9: numeric error;

 ! Display a very simple solution report;
 @WRITE( @NEWLINE( 1), '  A Simple Customized Solution Report', @NEWLINE( 1));
 @WRITE( 'Solution Status = ', ISTAT, @NEWLINE( 1));
 @WRITE( 'Objective= ', OBJ, @NEWLINE(1));
 @WRITE( 'Area= ', A, @NEWLINE(1));
! Can format output if desired;



 @WRITE( 'Cold Flow Mass= ', @FORMAT( CMASS, '12.3f'), @NEWLINE(1));

ENDCALC


